# Maintainer: David Runge <dave@sleepmap.de>
# Contributor: speps <speps at aur dot archlinux dot org>
# Contributor: Renato Budinich <rennabh at gmail dot com>

pkgname=wah-plugins
pkgver=0.1.0
pkgrel=6
pkgdesc="LADSPA Wah filter plugin"
arch=('loong64' 'x86_64')
url="https://kokkinizita.linuxaudio.org/linuxaudio/ladspa/"
license=('GPL2')
groups=('ladspa-plugins' 'pro-audio')
depends=('gcc-libs')
makedepends=('ladspa')
source=("https://kokkinizita.linuxaudio.org/linuxaudio/downloads/WAH-plugins-${pkgver}.tar.bz2")
sha512sums=('e3063115d2e635bee30ac4792d0c4f3536de79f25c153d790a1721c0f636636feeef5f7dd739c35443a68be058943ca29b4825dceef9bc56f862dbd7c7958ada')

prepare() {
  mv -v "WAH-plugins-${pkgver}" "${pkgname}-${pkgver}"
  
}

build() {
  cd "${pkgname}-${pkgver}"
  make
}

package() {
  cd "${pkgname}-${pkgver}"

  # plugin
  install -vDm 755 autowah.so "${pkgdir}/usr/lib/ladspa/autowah.so"
  # docs
  install -vDm 644 {AUTHORS,README} -t "${pkgdir}/usr/share/doc/${pkgname}/"
}

# vim:set ts=2 sw=2 et:
